
Highway

I thought that the old LCD games could be a base for games. I had HIGHWAY so this 
game is based on that LCD game.  Just set UDG's on when needed. 
Besides that it is just a fixed screen.

; Highway                                                       
; Game 54 in 1K hires for the ZX81                              
; Based on a LCD game with the same name                        
                                                                
? * TORNADO *                                                   
                                                                
           ORG  #4009                                           
           DUMP 49161                                           
                                                                
basic      LD   D,64              ; preset to copy disptab      
           JR   init0                                           
                                                                
           DEFB 236,212,28        ; The BASIC                   
           DEFB 126               ; fully placed over sysvar    
           DEFB 143,0,18          ; start to BASIC=#4009        
                                                                
eline      DEFW last              ; needed to load              
chadd      DEFW last-1                                          
xptr       DEFW 0                                               
stkbot     DEFW last              ; needed to load              
stkend     DEFW last              ; needed to load              
berg       DEFB 0                                               
mem        DEFW 0                                               
           DEFB 0                                               



                                                                
init1      JP   init              ; init can be anywhere        
                                                                
; all above reusable AFTER loading                              
                                                                
lastk      DEFB 255,255,255       ; used by ZX81                
margin     DEFB 55                ; used by ZX81                
nxtlin     DEFW basic             ; reusable after load         
                                                                
init0      XOR  A                 ; delay intrupts by           
           DEFB #16               ; LD D,64                     
flagx      DEFB 64                ; clever setting of flags     
                                                                
           LD   E,A               ; intruptcounter reset        
           EX   AF,AF'                                          
                                                                
taddr      DEFW 0                 ; used by ZX81,no hurting code
           LD   C,13              ; frames is set ok            
                                                                
frames     DEFW #DD*256+1         ; used by ZX81, clever IX set 
coprcc     LD   HL,hr             ; set IX                      
sposn      JR   init1                                           
cdflag     DEFB 64                ; used by zx81                
                                                                
hr         LD   HL,lowres+#8000   ; the lowres display          
           LD   BC,#269           ; minimum needed              
           LD   A,#1E                                           
           LD   I,A                                             
           LD   A,#FB                                           
           CALL #2B5                                            
                                                                
           LD   B,5               ; outline lowres with hires   
hr00       RET  M                 ; never true                  
           DJNZ hr00                                            
           NOP                                                  
                                                                
           LD   HL,screen-10      ; "start" of screendata       
           LD   DE,10             ; length of screenline        
           EXX                                                  
                                                                
           LD   D,lbuf/256+#80    ; set high byte lbuf          
                                                                
           LD   HL,#4000          ; display table               
           LD   BC,5              ; 5 layers of display         
                                                                
bwait      LD   B,12              ; line filler                 
bw         DJNZ bw                                              
                                                                
           LD   B,(HL)            ; get nr of display lines     
           INC  HL                                              
           LD   E,(HL)            ; get linebuffer to that lines
           INC  HL                                              
           JR   nline                                           
                                                                
dloop      EX   (SP),HL           ; delay                       
           EX   (SP),HL                                         
                                                                
           DEC  B                 ; alter bit 0 of B            
                                                                
           EXX                                                  
           JR   nline2                                          
                                                                
nline      EXX                                                  
           ADD  HL,DE             ; point to next dataline      



           LD   A,L                                             
           AND  127                                             
           JP   Z,repair          ; 0 and 128 gives displayerror
           DEFB 62                ; skip repair in 7 tstates    
repair     LD   L,(HL)            ; or get repair in 7 tstates  
                                                                
           LD   A,(HL)            ; filler                      
                                                                
nline2     LD   A,H                                             
           LD   I,A                                             
           LD   A,L                                             
           DEC  A                 ; JP (HL) will set R again ok 
           EXX                                                  
                                                                
           EX   DE,HL             ; get linebuffer              
           CALL #400A             ; LD R,A    JP (HL)           
           EX   DE,HL             ; save linebuffer             
           LD   A,B                                             
           RRCA                                                 
           JR   NC,dloop          ; each line 2x                
           DJNZ nline             ; do full block of lines      
           DEC  C                                               
           JR   NZ,bwait          ; do all block of lines       
                                                                
           CALL #292              ; back from intrupt           
           CALL #220                                            
           LD   IX,hr                                           
           JP   #2A4                                            
                                                                
eog        LD   HL,score-1        ; standard hiscore check      
           LD   DE,hiscore-1                                    
           LD   BC,6                                            
fihi       INC  HL                                              
           INC  DE                                              
           DEC  C                                               
           LD   A,(DE)                                          
           CP   (HL)                                            
           JR   Z,fihi                                          
           CALL C,#19F9           ; set hiscore                 
                                                                
start      LD   A,(lastk)         ; game over, wait for         
           SUB  %10111111         ; newline                     
           JR   NZ,start                                        
                                                                
           LD   A,31              ; set 3 lives                 
           LD   (lives),A                                       
                                                                
           LD   HL,score          ; reset score                 
           LD   B,5                                             
ersc       LD   (HL),28                                         
           INC  HL                                              
           DJNZ ersc                                            
                                                                
           LD   HL,lbuf                                         
           LD   B,5                                             
cls        CALL erlbuf            ; erase screen                
           DJNZ cls                                             
                                                                
nxtlive    LD   HL,lbuf+61                                      
           LD   (HL),64           ; erase left side lane        
           LD   L,lbuf*256/256+68                               
           LD   (carpos+1),HL     ; reset startposition         
                                                                
           CALL clseat            ; erase old hiker             



                                                                
playloop   LD   DE,lbuf+59        ; move streets down           
           LD   HL,lbuf+44                                      
           LD   BC,45                                           
           LDDR                                                 
                                                                
           INC  HL                                              
           CALL erlbuf            ; erase top line              
                                                                
           LD   B,3                                             
           CALL rnd                                             
           LD   HL,lbuf           ; set new obstacle on top     
           ADD  A,A                                             
           ADD  A,L                                             
           LD   L,A                                             
           LD   (HL),0            ; show obstacle               
           INC  HL                                              
           LD   (HL),0                                          
                                                                
dropcnt    LD   A,0                                             
           DEC  A                                               
           AND  15                                              
           LD   (dropcnt+1),A                                   
           JR   NZ,hitchcnt       ; each 16 steps show drop sign
           LD   (lbuf+9),A                                      
                                                                
hitchcnt   LD   A,20              ; counter hiker               
           DEC  A                                               
           JR   NZ,hikehigh                                     
                                                                
           LD   A,(seats)                                       
           OR   A                                               
           JR   NZ,hikehigh       ; hiker in car, no show       
                                                                
           INC  A                                               
           LD   (lbuf),A          ; show hitchhiker             
                                                                
           LD   B,20                                            
           CALL rnd                                             
           ADD  A,B                                             
                                                                
hikehigh   LD   (hitchcnt+1),A    ; new random start is set     
                                                                
doublp     PUSH AF                ; first no carry              
                                                                
carpos     LD   HL,0              ; erase car                   
           LD   (HL),64                                         
           INC  HL                                              
           LD   (HL),64                                         
           DEC  HL                                              
                                                                
           LD   A,%11011111                                     
           IN   A,(254)                                         
           LD   D,A               ; in port to D                
                                                                
           LD   BC,%1110000000000111                            
           LD   A,#A0             ; initialize ZXPAND joystick  
           OUT  (C),A                                           
           JR   zx2                                             
zx2        IN   A,(C)             ; read joystick udlrf...      
           RRA                    ; .udlrf..                    
           RRA                    ; ..udlrf.                    
           RRA                    ; ...udlrf                    
           RRA                    ; f...udlr                    



           CPL                                                  
           AND  3                 ; ......lr only               
           OR   D                 ; add keyboard input          
                                                                
           LD   C,L               ; save old X pos              
                                                                
           RRA                    ; P or right joystick         
           JR   C,left                                          
           INC  HL                ; move right                  
           INC  HL                                              
left       RRA                    ; O or left joystick          
           JR   C,valtest                                       
           DEC  HL                ; move left                   
           DEC  HL                                              
valtest    LD   A,L                                             
           SUB  lbuf*256/256+60   ; test left side move         
           JR   C,illmove         ; not out of screen left      
           PUSH HL                                              
           LD   HL,lbuf-15                                      
           JR   Z,sidetest        ; move to hiker allowed?      
           SUB  8                 ; test right side move        
           POP  HL                                              
           JR   C,nokey           ; not out of screen right     
           PUSH HL                                              
           LD   HL,lbuf-6                                       
           JR   NZ,illmove-1      ; legal move                  
                                                                
sidetest   LD   B,4               ; test hiker or sign shown    
t4         LD   DE,15                                           
           ADD  HL,DE                                           
           XOR  (HL)                                            
           DJNZ t4                                              
           POP  HL                                              
           JR   NZ,nokey          ; there is a hiker or sign    
           DEFB 62                                              
           POP  HL                                              
illmove    LD   L,C               ; illegal move                
nokey      LD   (carpos+1),HL     ; save car X pos              
                                                                
           LD   (HL),0            ; show car                    
           INC  HL                                              
           LD   (HL),0                                          
                                                                
           LD   HL,frames         ; delay                       
           LD   A,255-10                                        
           ADD  A,(HL)                                          
wfr        CP   (HL)                                            
           JR   NZ,wfr                                          
                                                                
           POP  AF                                              
           CCF                    ; change carry flag           
           JR   C,doublp          ; allow 2 moves per drop      
                                                                
           LD   A,C               ; get current position        
                                                                
gethitch   CP   lbuf*256/256+60                                 
           JR   NZ,droptest       ; no pick up                  
                                                                
           LD   A,(lbuf+45)                                     
           DEC  A                                               
           JR   NZ,droptest       ; hiker not yet at bottom     
                                                                
seat       LD   HL,seats          ; set hiker in the car        
           LD   A,8               ; first line of head          



sline      LD   B,5                                             
s1         LD   (HL),A                                          
           INC  HL                                              
           INC  HL                                              
           DJNZ s1                                              
           CP   20                                              
           LD   A,20              ; second line of head         
           JR   NZ,sline                                        
                                                                
droptest   LD   A,C                                             
           CP   lbuf*256/256+68   ; from above never on drop    
           JR   NZ,deadtest                                     
           LD   A,(lbuf+54)       ; get droppoint               
           LD   HL,seats                                        
           XOR  (HL)                                            
           XOR  8                 ; test on droppoint AND hiker 
           CALL Z,clseat          ; erase hiker from car        
           LD   HL,score+2        ; prepare 100 points          
           JR   Z,tenloop         ; from clseat also Z          
                                                                
deadtest   LD   HL,(carpos+1)                                   
           LD   A,L                                             
           SUB  15                                              
           LD   L,A                                             
           LD   A,(HL)                                          
           OR   A                 ; if line above visible, dead!
                                                                
           JR   Z,dead            ; dead                        
           CP   64                ; not aside means a point     
           LD   HL,score+4                                      
pl         JP   NZ,playloop                                     
                                                                
tenloop    INC  (HL)                                            
           LD   A,(HL)                                          
           CP   38                                              
           JR   NZ,pl                                           
           LD   (HL),28                                         
           DEC  HL                                              
           JR   tenloop                                         
                                                                
dead       LD   B,31              ; odd number will erase car   
deadshow   LD   A,64                                            
           LD   HL,(carpos+1)     ; flash the car               
           XOR  (HL)                                            
           LD   (HL),A                                          
           INC  HL                                              
           LD   (HL),A                                          
           LD   HL,frames         ; framedelay                  
           LD   A,(HL)                                          
deadfr     CP   (HL)                                            
           JR   Z,deadfr                                        
           DJNZ deadshow                                        
           LD   HL,lives                                        
           DEC  (HL)                                            
           LD   A,(HL)                                          
           CP   28                                              
           JP   Z,eog             ; game over                   
           JP   nxtlive           ; continue                    
                                                                
erlbuf     LD   A,64                                            
           LD   (HL),73           ; hide hiker and prevent score
           INC  HL                                              
                                                                
           XOR  (HL)              ; swap lines                  



           LD   (HL),A                                          
                                                                
           LD   C,3               ; 3 fields to hide            
           LD   A,64                                            
erline     INC  HL                                              
           LD   (HL),A                                          
           INC  HL                                              
           LD   (HL),A                                          
           DEC  C                                               
           JR   NZ,erline                                       
                                                                
           INC  HL                                              
           XOR  (HL)              ; swap lines                  
           LD   (HL),A                                          
           INC  HL                                              
           LD   A,64              ; hide dropzone               
           LD   (HL),A                                          
                                                                
           INC  HL                ; skip rest of buffer         
           INC  HL                ; can be shortened            
           INC  HL                ; but not necessary           
           INC  HL                                              
           INC  HL                                              
           INC  HL                                              
           RET                                                  
                                                                
clseat     LD   HL,seats          ; point to seat hiker         
           LD   C,2               ; 2 loops                     
clline     LD   B,5               ; 5 positions                 
clst       LD   (HL),0            ; erase seat                  
           INC  HL                                              
           INC  HL                                              
           DJNZ clst                                            
           DEC  C                                               
           JR   NZ,clline                                       
           RET                                                  
                                                                
rnd        LD   DE,0              ; default rnd routine         
           LD   HL,(frames)                                     
           ADD  HL,DE                                           
           DEC  HL                                              
           LD   A,H                                             
           AND  #1F                                             
           LD   H,A                                             
           LD   (rnd+1),HL                                      
           LD   A,(HL)                                          
frnd       SUB  B                                               
           JR   NC,frnd                                         
           ADC  A,B                                             
           RET                                                  
                                                                
x          EQU  101                                             
                                                                
lowres     DEFB 118                                             
score      DEFB 28,28,28,28,28,0                                
                                                                
           DEFB "H"+x,"I"+x,"G"+x,"H"+x,"W"+x,"A"+x,"Y"+x,0     
lives      DEFB 28,0                                            
hiscore    DEFB 28,28,28,33,32                                  
           DEFB 118                                             
                                                                
space      EQU  #426C-$           ; screen must be alligned     
                                                                
           DEFS space             ; free coding space           



                                                                
screen     DEFB 2,8,3,0,0,0,0,0,32,0                            
           DEFB 7,8,7,224,3,192,7,224,32,224                    
           DEFB 136,0,0,0,0,0,0,0 ; repairline for view         
           DEFB 2,8,3,192,1,128,10,16,32,224                    
           DEFB 5,8,1,64,1,128,7,224,32,64                      
                                                                
           DEFB 2,16,7,16,15,240,15,240,16,224                  
           DEFB 15,16,13,224,15,240,18,8,16,224                 
           DEFB 2,16,7,224,3,192,18,8,16,64                     
           DEFB 5,16,1,48,3,192,15,240,16,64                    
                                                                
           DEFB 14,32,14,0,31,248,15,240,8,112                  
           DEFB 46,32,27,4,54,108,20,8,8,80                     
           DEFB 52,32,15,248,31,248,34,164,8,112                
           DEFB 30,32,3,240,3,192,20,8,8,32                     
           DEFB 10,32,6,24,3,192,15,240,8,32                    
                                                                
           DEFB 4,64,30,0,127,254,1,0,4,0                       
           DEFB 8,0,0,0,0,0,0,0   ; repairline for view         
           DEFB 10,64,59,4,192,251,2,0,4,62                     
           DEFB 36,64,15,248,223,3,31,248,4,34                  
           DEFB 30,64,27,240,127,254,40,4,4,62                  
           DEFB 14,64,3,60,3,192,69,178,4,8                     
           DEFB 31,64,14,12,3,192,40,4,4,8                      
           DEFB 10,64,0,0,0,0,31,248,4,8                        
                                                                
           DEFB 63,252,63,252,63,252,63,252,63,252              
           DEFB 66,66,66,66,66,66,66,66,66,66                   
           DEFB 65,130,65,130,65,130,65,130,65,130              
seats      DEFB 8,16,8,16,8,16,8,16,8,16                        
           DEFB 20,40,20,40,20,40,20,40,20,40                   
           DEFB 126,126,126,126,126,126,126,126,126,126         
           DEFB 129               ; repair line for view        
           DEFB 66,66,66,66,66,66,66,66,66,66                   
                                                                
lbuf       DEFB 1,9,0,0,0,0,0,0,9,0,64,64,64,64                 
           RET                                                  
                                                                
init       LD   SP,#4400                                        
           LD   HL,disptab                                      
           LD   B,E                                             
           LDIR                   ; copy disptab over sysvar    
           LD   HL,lbuf                                         
           LD   DE,init                                         
           LD   C,4*15                                          
           SCF                                                  
           JP   start-3                                         
                                                                
disptab    DEFB 8,lbuf*256/256                                  
           DEFB 8,lbuf*256/256+15                               
           DEFB 10,lbuf*256/256+30                              
           DEFB 14,lbuf*256/256+45                              
           DEFB 14,lbuf*256/256+60                              
                                                                
L400A      LD   R,A               ; the display call for each   
           JP   (HL)              ; line copies to #400A        
                                                                
vars       DEFB 128                                             
?                                                               
last       EQU  $                                               


